

# Calcul du paiement du solde des congés à la clôture

Le principe est de réaliser le calcul de la monétarisation du solde des droits restants en appliquant le comparatif des 10%.

Cette codification ne peut être livrée en standard car elle dépend des choix du clients (accords internes temps de travail) et des modèles de congés en place.

Le référentiel utilisé pour cet exemple est le suivant :

- Un modèle de congés annuels nommé CONGESANNUEL
- Un modèle de congés d'ancienneté déclaré comme congé annuel supplémentaire nommé ANC66
- Paiement des soldes des congés sur 2 années antérieures plus l'année en cours.

## 1°) Définition de la rubrique libre réalisant les calculs intermédiaires.

Ajouter une rubrique libre marquée "toujours calculée" et les formules suivantes.

- ALIAS : **CALCULS\_ICCP**
- LABEL : **Calcul 10% sur clôture.**

Cette rubrique permettra d'alimenter la rubrique **CP\_PAIE** ou une rubrique de paie de votre choix.

### a) Définition de l'année de référence.

Formule **ANNEE\_REF** - Année de référence.

Elle permet de disposer de l'année à utiliser pour identifier la période d'acquisition des congés actuellement en cours. Elle est dépendante de la codification des congés en place chez le client. Dans notre exemple l'acquisition est déclarée de Juin N-1 à Mai N.

Elle est donc à adapter si les périodes d'acquisitions en place sont différentes.

```
SI( CONSTANTE( GENERAL. MOIS_PAYE ) < 6)  
alors( CONSTANTE( GENERAL. ANNEE_PAYE ) - 1)
```

```
sinon( CONSTANTE( GENERAL. ANNEE_PAYE) )
```

Valider la formule et la rubrique puis recharger la rubrique.

## b) Solde des droits sur la période N - 2.

Formule **SOLDE\_N\_2** - Solde N - 2.

- On utilise le mot clé **acquisition** avec le paramètre **5** qui permet de récupérer le solde des droits avant clôture.
- On travaille avec la formule **ANNEE\_REF** - 2.

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF]-2; 5) +  
acquisition( ANC66; [ _CALCULS_ICCP. ANNEE_REF]-2; 5)
```

## c) Solde des droits sur la période N - 1.

Formule **SOLDE\_N\_1** - Solde N - 1.

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF]-1; 5) +  
acquisition( ANC66; [ _CALCULS_ICCP. ANNEE_REF]-1; 5)
```

## d) Solde des droits sur la période en cours d'acquisition.

Formule **SOLDE\_IN** - Solde des droits en cours d'acquisition.

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF]; 5) +  
acquisition( ANC66; [ _CALCULS_ICCP. ANNEE_REF]; 5)
```

## e) Droits acquis sur la période N - 2.

Formule **ACQUIS\_N\_2** - Droits acquis N - 2

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF] - 2; 1)
```

## f) Droits acquis sur la période N - 1.

Formule **ACQUIS\_N\_1** - Droits acquis N - 1

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF] - 1; 1)
```

## g) Droits acquis sur la période en cours d'acquisition.

Formule **ACQUIS\_IN** - Droits acquis sur N

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF]; 1)
```

## h) Assiette sur la période N - 2.

Formule **ASSIETTE\_N\_2** - Assiette N - 2.

La récupération de l'assiette doit impérativement être réalisée avec le mot clé **acquisition** avec le paramètre **6**. Si vous utilisez une lecture de l'historique de paie vous ne disposez pas des ajustements réalisés individuellement sur chaque fiche contractuelle.

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF] -2; 6)
```

## i) Assiette sur la période N - 1.

Formule **ASSIETTE\_N\_1** - Assiette N - 1.

```
acquisition( CONGESANNUEL; [ _CALCULS_ICCP. ANNEE_REF] -1; 6)
```

## j) Assiette sur la période en cours.

Ce calcul nécessite l'utilisation de deux formules.

- La première permet de calculer l'assiette de la période courante sans déclencher une référence circulaire.
- La seconde permet de cumuler l'assiette des mois antérieurs et le montant de la formule précédente.

Formule **ASSIETTE\_PAYE** - Assiette du mois courant.

On prend le montant du BRUT avec exclusion de la rubrique CP\_PAIE (ou celle de votre choix rentrant dans le calcul du brut) auquel on ajoute le montant des éléments de B\_CP\_ASSIETTE hors BRUT.

Si vous utilisez une rubrique de paie utilisateur, pensez à modifier cette formule en remplaçant CP\_PAIE par votre rubrique.

```
EXCLURE( [ BRUT. MONTANT]; [ CP_PAIE. MONTANT] ) +  
EXCLURE( [ B_CP_ASSIETTE. MONTANT]; [ BRUT. MONTANT] )
```

Valider la formule et la rubrique puis recharger la rubrique.

Formule **ASSIETTE\_IN** - Assiette de la période d'acquisition courante

```
acquisition(CONGESANNUEL; [_CALCULS_ICCP. ANNEE_REF]; 6) +  
[_CALCULS_ICCP. ASSIETTE_PAYE]
```

Valider la formule et la rubrique puis recharger la rubrique.

## k) Valorisation des soldes N - 2.

Formule **CALCUL\_N\_2** - Calcul du paiement des CP N - 2.

```
SI([_CALCULS_ICCP. ACQUIS_N_2]>0)  
alors(  
  si([_CALCULS_ICCP. SOLDE_N_2]>0)  
  alors(  
    [_CALCULS_ICCP. ASSIETTE_N_2] * 0.1 / [_CALCULS_ICCP. ACQUIS_N_2] *  
    [_CALCULS_ICCP. SOLDE_N_2]  
  )  
  sinon(0)  
)  
sinon(0)
```

## l) Valorisation des soldes N - 1.

Formule **CALCUL\_N\_1** - Calcul du paiement des CP N - 1.

```

SI([_CALCULS_ICCP.ACQUIS_N_1]>0)
alors(
  si([_CALCULS_ICCP.SOLDE_N_1]>0)
  alors(
    [_CALCULS_ICCP.ASSIETTE_N_1] * 0.1 / [_CALCULS_ICCP.ACQUIS_N_1] *
[_CALCULS_ICCP.SOLDE_N_1]
  )
  sinon(0)
)
sinon(0)

```

## m) Valorisation des soldes N.

Formule **CALCUL\_IN** - Calcul du paiement des CP sur la période courante.

```

SI([_CALCULS_ICCP.ACQUIS_IN]>0)
alors(
  si([_CALCULS_ICCP.SOLDE_IN]>0)
  alors(
    [_CALCULS_ICCP.ASSIETTE_IN] * 0.1 / [_CALCULS_ICCP.ACQUIS_IN] * [_CALCULS_ICCP.SOLDE_IN]
  )
  sinon(0)
)
sinon(0)

```

Valider la formule et la rubrique puis recharger la rubrique.

## n) Total final.

Enfin on cumule les calculs intermédiaires.

Formule **TOTAL** - Total cumulé.

Le résultat de cette formule est pilotée par les conditions suivantes :

- Le contrat est clos.
- La clôture n'est pas un changement de situation.
- Le contrat ne fait pas l'objet du paiement des 10% CDD.

```

DECLARER( PRENDRE)
AFFECTER( PRENDRE;

```

```

/* contrat clos */
SI( CONSTANTE( CONTRAT. CLOTURE) =1)
ALORS( 1)
SINON( 0)
*
/* Pas de paiement des 10% CDD */
SI( [ I_ICP. MONTANT] >0)
ALORS( 0)
SINON( 1)
*
/* Pas sur changement de situation */
SI ( CONSTANTE( CONTRAT. TYPECLOTURE) <> 1)
ALORS( 1)
SINON( 0)
)

SI( : PRENDRE=1)
ALORS(
  [ _CALCULS_ICCP. CALCUL_N_2] +
  [ _CALCULS_ICCP. CALCUL_N_1] +
  [ _CALCULS_ICCP. CALCUL_IN]
)
SINON( 0)

```

## 2°) Intégration dans le calcul des paies.

- La rubrique de paie permettant d'intégrer ce calcul au BRUT lors de la clôture doit disposer des paramètres suivants :
- Toujours valorisée.
- Multiples instances autorisé.
- Rubrique d'écrasement.
- Imprimer sur le bulletin sous contrôle de valorisation du MONTANT.
- Saisissable au niveau des variables de paie.

Rubrique en Point <input type="checkbox"/>	Rubrique en Heure <input type="checkbox"/>	Toujours Valorisée <input checked="" type="checkbox"/>
Unicité à l'édition <input type="checkbox"/>	Multipl. instances autorisé <input checked="" type="checkbox"/>	Rubrique d'écrasement <input checked="" type="checkbox"/>
Désactiver dans les bulletins de régularisation <input type="checkbox"/>		Non saisissable <input type="checkbox"/>
Imprimer sur le bulletin <input checked="" type="checkbox"/>	sous contrôle de valorisation <input checked="" type="checkbox"/>	sur les formules : <input type="checkbox"/> BASE <input type="checkbox"/> TAUX
Via une condition utilisateur <input type="checkbox"/>		<input type="button" value="Définir..."/> <input checked="" type="checkbox"/> MONTANT
A éditer dans le bulletin annexe <input type="checkbox"/> Libellé de substitution : <input type="text"/>		
<input type="checkbox"/> Ne pas appliquer l'historisation conditionnelle		
<b>Fréquence de calcul</b>		
<input type="button" value="Jan."/> <input type="button" value="Fev."/> <input type="button" value="Mars"/> <input type="button" value="Avr."/> <input type="button" value="Mai"/> <input type="button" value="Juin"/> <input type="button" value="Jui."/> <input type="button" value="Aout"/> <input type="button" value="Sep."/> <input type="button" value="Oct."/> <input type="button" value="Nov."/> <input type="button" value="Dec."/>		
<b>Niveaux d'utilisation</b>		
Convention <input type="checkbox"/>	Grille de Convention <input type="checkbox"/>	Usages établissement <input type="checkbox"/>
Contrat <input type="checkbox"/>	Saisie de Paye <input checked="" type="checkbox"/>	

Dans la formule BASE de cette rubrique, il suffit d'appeler le résultat du calcul de la rubrique libre.

```
[ _CALCULS_ICCP. TOTAL ]
```

Cette codification est fonctionnelle et a déjà été mise en place chez plusieurs clients. Néanmoins, il est nécessaire de prendre en compte les spécificités propres à chacun et adapter son fonctionnement. Toute mise en place doit être testée et faire l'objet d'une recette avec le client.

Revision #5

Created 25 May 2026 10:33:17 by Christophe.F

Updated 25 May 2026 12:47:27 by Christophe.F